Health Care Operations Management: Facility Location

Amir A. Nasrollahzadeh

Clemson University

snasrol@g.clemson.edu

Fall 2018

Overview

Introduction

The location of facilities is critical in both industry and in health care. In health care, the implications of poor location decisions extend well beyond cost and customer service considerations.



Introduction

- Too many facilities (never happens!) lead to heavy capital and budgeting costs.
- Too few facilities increase mortality and morbidity (disease).
- Even for just the right number of facilities, poorly located ones would cause accessibility and availability issues.
    - What if a small portion of population are located far from the health care center?
    - What may go wrong with locating a health care facility near a densely populated area?

Locating a health care facility

Coverage

In lots of health care applications, **coverage** is used to determine if the distance (travel time) between demand and service locations is less than or equal to some externally identified threshold.

- For example, the U.S. National Fire Protection Association suggests a target of 90% of calls responded to within 4 minutes for the first response to an urgent call, followed by an Advanced Life Support response within 8 minutes (NFPA 2010).

Coverage is often the motivation for different objective functions in health care facility location problems.

Locating a health care facility

What if a small portion of population are located far from the health care center?

Coverage drops for scarcely populated areas but densely populated areas may get a boost in coverage. A small portion of population usually represents a small portion of demand!

What may go wrong with locating a health care facility near densely populated areas?

Distance to the health care center definitely drops. However, densely populated areas have traffic issues which may have a negative effect on coverage in terms of travel time.

Facility location models

Discrete models assume

- Demand is represented by a limited number of discrete points
- A finite set of candidate locations exists for building a facility

Continuous models assume

- Demand is distributed continuously across a region
- Facilities can generally be located anywhere

Set covering model

Suppose that the state is divided into five zones for demand and facility locations. Each demand zone must be covered by at least one facility. A demand zone is covered by a facility if they are located within 40 miles of each other. Find the optimal location of health care facilities such that the building cost is minimized. How many health care centers are needed? The building cost in each zone and the distance between them is given.



Set covering model

Data and parameters

- Demand zones: 1, 2, 3, 4, 5.
- Candidate facility locations: 1, 2, 3, 4, 5.
- Fixed building costs: $(c_1, c_2, c_3, c_4, c_5) = (30, 15, 40, 30, 20)$.
- Distance:

d_{ij}	1	2	3	4	5
1	0	20	30	40	50
2	20	0	20	45	40
3	30	20	0	30	30
4	40	45	30	0	15
5	50	40	30	15	0

$\xrightarrow{\text{coverage?}}$

a_{ij}	1	2	3	4	5
1	1	1	1	1	0
2	1	1	1	0	1
3	1	1	1	1	1
4	1	0	1	1	1
5	0	1	1	1	1

Set covering model

Decision variables

The objective is to find optimal locations of facilities such that the building cost is minimized and each demand zone is covered. We know that once a facility is built, the zone it was built in is covered. We can safely say that every demand zone is covered if a facility is built in each zone. Do we have to build a facility in each zone?

$$x_j = \begin{cases} 1 & \text{if a facility is built in zone } j, \\ 0 & \text{otherwise,} \end{cases} \qquad \forall j \in F.$$

For example, $x_1 = 1, x_2 = 0$ means that a facility is built in zone 1 but not in zone 2.

Set covering model

Objective function

The objective is to find optimal locations of facilities such that the building cost is minimized and each demand zone is covered. The decision variables determine if a facility is located in each zone. We have the building cost of each facility as a parameter. We have to minimize the total cost of building facilities.

$$\min c_1x_1 + \boxed{c_2x_2} + c_3x_3 + c_4x_4 + c_5x_5$$

If $x_2 = 1$, a facility is built in zone 2, and the cost c_2 will be included in total cost. If $x_2 = 0$, c_2 will not be included in the total cost.

Set covering model

Constraints

The objective is to find optimal locations of facilities such that the building cost is minimized and each demand zone is covered. Based on the coverage matrix given as part of the parameters, we know whether a facility at zone j can cover demand zone i. We must insure that each demand zone is covered by at least one facility.

$$\begin{array}{ll} \text{Demand zone 1:} & 1x_1 + 1x_2 + 1x_3 + 1x_4 + 0x_5 \geq 1, \\ \vdots & \vdots \\ \text{Demand zone 5:} & 0x_1 + 1x_2 + \boxed{1x_3} + 1x_4 + 1x_5 \geq 1, \end{array}$$

Coverage matrix $[a_{i,j}]$ tells us that demand zone 5 can be covered by a facility in zone 3. If we build a facility in one of $x_2, \ldots, x_5$, meaning at least one of $x_2, \ldots, x_5$ equals 1, demand zone 5 is covered.

Set covering model: Formal formulation

Parameters

Demand zones: $D := \{1, 2, \dots, d\}$

Facility zones: $F := \{1, 2, \dots, f\}$,

Building cost: $c_j, \qquad \forall j \in F$,

Coverage: $a_{ij} = \begin{cases} 1 & \text{if demand } i \text{ is covered by facility } j, \\ 0 & \text{otherwise}, \qquad \forall j \in F, \forall i \in D. \end{cases}$

Model

$$
\begin{array}{ll}
\min & \sum_{j \in F} c_j x_j, \\
s.t. & \\
1: & \sum_{j \in F} a_{ij} x_j \geq 1, \quad \forall i \in D, \\
 & \\
2: & x_j \in \{0, 1\} \qquad \forall j \in F.
\end{array}
$$

Set covering model: AMPL

Model file: set-covering.mod

```
##parameters##
param D; # number of demand zones
param F; # number of facility locations
param c{j in 1..F}; # cost of building at location j
param a{i in 1..D, j in 1..F}; # whether zone i is covered by zone j
##variables##
var x{j in 1..F} binary; # whether a facility is built in zone j
##objective##
minimize cost: sum{j in 1..F} c[j] * x[j];
##constraints##
subject to demand{i in 1..D}: sum{j in 1..F} a[i,j] * x[j]>=1;
```

Set covering model: AMPL

data file: set-covering.dat

```
##parameters##
param D:=5; # number of demand zones
param F:=5; # number of facility locations
# cost of building at location j
param c:= 1 30 2 15 3 40 4 30 5 20;
# whether zone i is covered by zone j
param a:  1  2  3  4  5:=
       1  1  1  1  1  0
       2  1  1  1  0  1
       3  1  1  1  1  1
       4  1  0  1  1  1
       5  0  1  1  1  1;
```

Set covering model: AMPL

run in AMPL

```
model set-covering.mod;
data set-covering.dat;
option solver cplex;
solve;
display x;
display cost;
```

Set covering model

- Q1: Suppose that the fixed building cost is the same for all candidate facility location. What change is needed to minimize the number of facilities required to cover all demand zones?
- Q2: What would be the effect of increasing the right hand side of constraint-1?
- Q3: In this formulation, all demands are treated equally in terms of coverage. Is there any motivation to differentiate demand zones?

Maximal covering model

SC state is divided into five zones for demand and facility locations. Demand in each zone is denoted by *h*. A demand zone is covered if it is located within 40 miles of a facility. Find the optimal locations of health care facilities such that coverage is maximized. The building cost is the same for all facility locations but we only have the budget to build two facilities.



Maximal covering model

Data and parameters

- Demand zones: 1, 2, 3, 4, 5.
- Candidate facility locations: 1, 2, 3, 4, 5.
- Fixed demands: $(h_1, h_2, h_3, h_4, h_5) = (30, 15, 40, 30, 20)$.
- Number of facilities: 2
- Distance:

d_{ij}	1	2	3	4	5
1	0	20	30	40	50
2	20	0	20	45	40
3	30	20	0	30	30
4	40	45	30	0	15
5	50	40	30	15	0

$\xrightarrow{\text{coverage?}}$

a_{ij}	1	2	3	4	5
1	1	1	1	1	0
2	1	1	1	0	1
3	1	1	1	1	1
4	1	0	1	1	1
5	0	1	1	1	1

Maximal covering model

Decision variables

The goal is to find optimal locations of facilities such that demand coverage is maximized. Similar to set covering model, we need to know if a facility is located at a specific node. Therefore, one set of decision variables are

$$x_j = \begin{cases} 1 & \text{if a facility is built in zone } j, \\ 0 & \text{otherwise,} \end{cases} \qquad \forall j \in F.$$

However, we can not determine how much demand is covered by relying only on x_j variables. Covered demand must be known in order to evaluate the coverage in the objective function.

$$y_i = \begin{cases} 1 & \text{if demand zone } i \text{ is covered,} \\ 0 & \text{otherwise,} \end{cases} \qquad \forall i \in D.$$

Maximal covering model

Objective function

The goal is to find optimal locations of facilities such that demand coverage is maximized. Since we defined y_i variables to identify which demand zone is covered, evaluating total covered demand is straightforward.

$$\max h_1 y_1 + \boxed{h_2 y_2} + h_3 y_3 + h_4 y_4 + h_5 y_5$$

If $y_2 = 1$, demand zone 2 is covered by a facility and its demand, h_2 will be included in total covered demand.

Maximal covering model

Constraints

Only one of the constraints is explicitly identified in the problem statement. We only have the budget to build two facilities. We know x_j variables determine what locations are chosen for building a facility. Therefore,

Two facilities: $$x_1 + x_2 + x_3 + x_4 + x_5 \leq 2$$

This constraint will not allow three or more x_j variables to equal 1. At most two facilities will be built.

Maximal covering model

Constraints (continued)

The problem statement does not explicitly identify a coverage constraint like the set covering model. However, is it possible to set $y_1 = 1$ and only build a facility in zone 5, i.e., $x_5 = 1$?

$$\begin{array}{ll} \text{Demand zone 1:} & 1x_1 + 1x_2 + 1x_3 + 1x_4 + 0x_5 \geq y_1, \\ \vdots & \vdots \\ \text{Demand zone 5:} & \boxed{0x_1 + 1x_2 + 1x_3 + 1x_4 + 1x_5 \geq y_5,} \end{array}$$

Remember the set covering constraints. The difference here is that the right hand side has changed from 1 to y_i. This change means that **if** demand zone 1 is covered, i.e., $y_5 = 1$, then one of x_2, x_3, x_4, and x_5 should equal 1.

Maximal covering model: Formal formulation

Parameters

Demand zones: $D := \{1, 2, \ldots, d\}$

Facility zones: $F := \{1, 2, \ldots, f\}$,

Demand: $h_i, \quad \forall i \in D$,

\# of facilities: P,

Coverage: $a_{ij} = \begin{cases} 1 & \text{if demand } i \text{ is covered by facility } j, \\ 0 & \text{otherwise}, \qquad \forall j \in F, \forall i \in D. \end{cases}$

Model

$$\begin{aligned} \max \quad & \textstyle\sum_{i \in D} h_i \, y_i, \\ s.t.\ 1: \quad & \textstyle\sum_{j \in F} x_j \leq P, \\ 2: \quad & \textstyle\sum_{j \in F} a_{ij} x_j \geq y_i, \quad \forall i \in D, \\ 3: \quad & x_j \in \{0, 1\} \qquad \forall j \in F, \\ 4: \quad & y_i \in \{0, 1\} \qquad \forall i \in D. \end{aligned}$$

Maximal covering model: AMPL

Model file: max-covering.mod

```
param D; # number of demand zones
param F; # number of facility locations
param P; # max number of facilities
param h{i in 1..D}; # demand at location i
param a{i in 1..D, j in 1..F}; # whether zone i is covered by zone j
##variables##
var x{j in 1..F} binary; # whether a facility is built in zone j
var y{i in 1..D} binary; # whether demand zone i is covered
##objective##
maximize coverage: sum{i in 1..D} h[i] * y[i];
##constraints##
subject to facilities: sum{j in 1..F } x[j]<=P;
subject to demand{i in 1..D}: sum{j in 1..F} a[i,j] * x[j]>=y[i];
```

Maximal covering model: AMPL

data file: max-covering.dat

```
##parameters##
param D:=5; # number of demand zones
param F:=5; # number of facility locations
param P:=2; # max number of facilities
# demand at location i
param h:= 1 30 2 15 3 40 4 30 5 20;
# whether zone i is covered by zone j
param a:  1  2  3  4  5:=
       1  1  1  1  1  0
       2  1  1  1  0  1
       3  1  1  1  1  1
       4  1  0  1  1  1
       5  0  1  1  1  1;
```

Maximal covering model: AMPL

run in AMPL

```
model max-covering.mod;
data max-covering.dat;
option solver cplex;
solve;
display x;
display y;
display coverage;
```

Maximal covering model

- Q1: According to constraint-2 in the maximal covering model, when is it possible to say that demand zone-1 is covered?
- Q2: What is the difference between constraint-1 in the set covering model and constraint-2 in the maximal covering model?
- Q3: Is there a balanced approach to locate facilities such that the distance to all demand is minimum?

P-median model

SC state is divided into five zones for demand and facility locations. Demand in each zone is denoted by h. Find the optimal locations of health care facilities such that the total demand-weighted distance between all demands and facilities is minimized. The building cost is the same for all facilities but we only have the budget to build on two locations.



P-median model

Data and parameters

- Demand zones: 1, 2, 3, 4, 5.
- Candidate facility locations: 1, 2, 3, 4, 5.
- Fixed demands: $(h_1, h_2, h_3, h_4, h_5) = (30, 15, 40, 30, 20)$.
- Number of facilities: 2
- Distance:

d_{ij}	1	2	3	4	5
1	0	20	30	40	50
2	20	0	20	45	40
3	30	20	0	30	30
4	40	45	30	0	15
5	50	40	30	15	0

P-median model

Decision variables

The problem statement does not mention coverage. The goal is still to find the optimal locations of facilities such that the total demand-weighted distance between demand zones and facility locations is minimized. Therefore,

$$x_j = \begin{cases} 1 & \text{if a facility is built in zone } j, \\ 0 & \text{otherwise,} \end{cases} \qquad \forall j \in F.$$

In order to write the objective function, we need to know which demand zones are assigned to each facility so that we can evaluate the distance between demands and facilities.

$$y_{ij} = \begin{cases} 1 & \text{if demand zone } i \text{ is assigned to facility } j, \\ 0 & \text{otherwise,} \end{cases} \qquad \forall i \in D, \forall j \in F.$$

P-median model

Objective function

The objective is to place facility locations such that the total demand-weighted distance between demand zones and facility locations is minimized. The first step is to evaluate the demand-weighted distance.

- Demand-weighted distance:

$$h_1(d_{11}+d_{12}+\ldots+d_{15})+h_2(d_{21}+d_{22}+\ldots+d_{25})+\ldots+h_5(d_{51}+d_{52}+\ldots+d_{55}).$$

- Objective:

$$\min h_1(d_{11}y_{11}+\ldots+d_{15}y_{15})+\ldots+\boxed{h_5(d_{51}y_{51}+\ldots+d_{55}y_{55})}$$

If $y_{51}=1$, demand zone 5 is assigned to facility 1 and its demand, h_5 will be satisfied over a distance of d_{51}.

P-median model

Constraints

The only explicit constraint identified in problem statement is the limited budget to build two facilities. We know x_j variables determine facility locations. Therefore,

Two facilities: $$x_1 + x_2 + x_3 + x_4 + x_5 \leq 2$$

This constraint will not allow three or more x_j variables to equal 1. At most two facilities will be built.

P-median model

Constraints (continued)

Let's see if there are any implied constraints in the problem statement. A usually good starting point is to explore the relationship between decision variables. Is it possible to set $y_{21} = 1$ and $x_1 = 0$?

$$\begin{array}{ll} \text{Facility 1:} & y_{11} \leq x_1, y_{21} \leq x_1, \ldots, y_{51} \leq x_1, \\ \vdots & \vdots \\ \text{Facility 5:} & \boxed{y_{15} \leq x_5}, y_{25} \leq x_5, \ldots, y_{55} \leq x_5, \end{array}$$

The demand in zone 1 can only be covered by a facility at location 5, i.e., $y_{15} = 1$, if a facility is already built at location 5, i.e., $x_5 = 1$. If $x_5 = 0$, no facility is built in location 5, then no demand can be covered in this location and all $y_{15}, y_{25}, y_{35}, y_{45}, y_{55}$ equal 0.

P-median model

Constraints (continued)

There is another implied constraint in this problem statement. Is it a good idea to assign demand zone 1 to facility locations 2 and 3, i.e., $y_{12} = 1$ and $y_{13} = 1$? Notice that if one demand zone is assigned to two or more facilities, its distance to those facilities will be included in the objective function.

$$\begin{aligned} \text{Demand zone 1:} \quad & y_{11} + y_{12} + y_{13} + y_{14} + y_{15} = 1, \\ \vdots \quad & \qquad \vdots \\ \text{Demand zone 5:} \quad & \boxed{y_{51} + y_{52} + y_{53} + y_{54} + y_{55} = 1}, \end{aligned}$$

Demand zone 5 is only assigned to one of the facilities because exactly one of $y_{51}, \ldots, y_{55}$ can equal 1.

P-median model: Formal formulation

Parameters

$$
\begin{array}{ll}
\text{Demand zones:} & D := \{1, 2, \ldots, d\} \\
\text{Facility zones:} & F := \{1, 2, \ldots, f\}, \\
\text{Demand:} & h_i, \qquad \forall i \in D, \\
\text{\# of facilities:} & P, \\
\text{Distance:} & d_{ij} \qquad \forall i \in D, \forall j \in F.
\end{array}
$$

Model

$$
\begin{array}{rll}
\min & \sum_{i \in D} \sum_{j \in F} h_i d_{ij} y_{ij}, & \\
s.t.\ 1: & \sum_{j \in F} x_j \leq P, & \\
2: & y_{ij} \leq x_j, & \forall i \in D, \forall j \in F, \\
3: & \sum_{j \in F} y_{ij} = 1, & \forall i \in D, \\
4: & x_j \in \{0, 1\} & \forall j \in F, \\
5: & y_{ij} \in \{0, 1\} & \forall i \in D, \forall j \in F.
\end{array}
$$

Accessibility

Definition

The ability of patients or clients to reach the health care facility or, in the case of emergency services, the ability of the health care providers to reach patients.

- The EMS (Emergency Medical Services) Act of 1973 stipulated that 95% of service requests had to be served within 30 minutes in a rural area and within 10 minutes in an urban area.

Adaptability

Definition

Location decisions must be robust with respect to uncertain future conditions, particularly for facilities such as hospitals that are difficult if not impossible to relocate as conditions change.

- Scenario planning is frequently used to handle future uncertainty. A number of future conditions are defined and plans are developed that do well in all (or most) scenarios.
- Designing a robust system often entails compromises. The best compromise plan may not be optimal under any scenario but will do well across all scenarios.

Availability

Definition

In contrast to adaptability which considered long-term uncertainty, availability addresses very short-term changes in the condition of the system that result from facilities being busy.

- Such models are most applicable to emergency service systems (ambulances) in which a vehicle may be busy serving one demand at the time it is needed to respond to another emergency.

Class work

- 1: P-median. Code the p-median facility location model in AMPL. Run and report the results. Use data given in the problem statement presented in class.
  Modify the problem such that the objective function minimizes weighted average of the distance between demand zones and facility locations. Compare the results.

Class work

- 2: Consider availability. In all the three models we discussed, availability is an issue. A health care facility such as an emergency room might be busy at any given time. One way to address this problem is to ensure that each demand zone is covered by multiple facilities. The chance that two emergency rooms are busy at the same time is considerably less. Which model (set covering, maximal covering, and p-model) is able to cover demand zones with multiple facilities? How? Write the model and modify its elements so that each demand zone is covered at least by two facilities.

Class work

- 3: Consider accessibility. Some times this concept is defined in terms of coverage. For this problem, assume a demand zone is covered if it is located within 20 miles of a facility. Modify the p-median problem to minimize the distance between all demands and facilities such that each demand zone is provided with an accessible health care facility.

Project

- Consider adaptability. Travel times in dry and wet seasons may differ substantially. A demand zone which is covered by a facility may not be accessible in heavy snowfall. Suppose that the d_{ij} matrix given in previous slides denotes travel times between demand zones and facility locations. In wet conditions, it takes $1.5 \times d_{ij}$ to travel between zone i and j. Denote dry travel times as $d_{ij,\text{dry}}$, and wet travel times as $d_{ij,\text{wet}}$. Any given day may be a dry day with probability $\frac{2}{3}$ (i.e., probability of wet day$=\frac{1}{3}$). Find optimal locations of facilities such that the **expected demand-weighted travel times** between all demand zones and facility locations is minimized in **wet and dry** conditions.

Project: Hints

- This is called robust optimization by scenario planning. Travel times differ under different scenarios (dry and wet conditions). The model to be modified is the p-median model.
- The decision variable describing the assignment of demand i to facility j should reflect the weather condition. Maybe $y_{ij,\text{dry}}$ and $y_{ij,\text{wet}}$.
- The objective is to minimize the **expected** demand-weighted travel times. The probability of each scenario is given. The final objective function should take the expectation of dry and wet objectives.
- In coding the model, you need to input three dimensional data in AMPL.

Sources:

- Brandeau, M.L., Sainfort, F. and Pierskalla, W.P. eds., 2004. Operations research and health care: a handbook of methods and applications (Vol. 70). Springer Science & Business Media.